
pro read_swath_l1_airs, pattern, numfp, numline, $
                        tai, lat, lon, rad, solzen, $
                        scanang=scanang, $
                        sun_glint_distance=sun_glint_distance, $
                        CalFlag=CalFlag, $
                        freq=freq, topog=topog, state=state

   filename = findfile ( pattern, count = cnt )
   if cnt ne 1 then begin
      print, 'findfile did not return exactly one file ', cnt, pattern
      return
   endif

   print, 'read_swath_l1_airs: ', filename[0]

   FAIL = -1
   SWATH_L1AIRS = 'L1B_AIRS_Science'
   cc_level1 = 0

   fid=EOS_SW_OPEN(filename[0], /READ)
   if fid eq FAIL then begin
      return
   endif

   SWid=EOS_SW_ATTACH(fid, SWATH_L1AIRS)
   if SWid eq FAIL then begin
      SWATH_L1AIRS = 'L2_Standard_cloud-cleared_radiance_product'
      SWid=EOS_SW_ATTACH(fid, SWATH_L1AIRS)
      cc_level1 = 1
   endif

   ret = EOS_SW_READFIELD(SWid,"Time",tai)
   sz=size(tai)
   numfp=sz[1]
   numline=sz[2]

   ret = EOS_SW_READFIELD(SWid,"Latitude",lat)
   ret = EOS_SW_READFIELD(SWid,"Longitude",lon)
   ret = EOS_SW_READFIELD(SWid,"solzen",solzen)
   ret = EOS_SW_READFIELD(SWid,"radiances",rad)

   if arg_present(scanang) then begin
      print, "reading the scan angle"
      ret = EOS_SW_READFIELD(SWid,"scanang",scanang)
   endif

   if arg_present(sun_glint_distance) then begin



      print, "reading the sun glint distance"
      ret = EOS_SW_READFIELD(SWid,"sun_glint_distance",sun_glint_distance)
   endif

   if arg_present(CalFlag) then begin
      print, "reading the CalFlag"
      ret = EOS_SW_READFIELD(SWid,"CalFlag",CalFlag)
   endif

   if arg_present(freq) then begin
      print, "reading the frequencies"
      if cc_level1 eq 0 then begin
         ret = EOS_SW_READFIELD(SWid,"spectral_freq", freq )
      endif else begin
         ret = EOS_SW_READFIELD(SWid,"freq", freq )
      endelse
   endif

   if arg_present(state) then begin
      print, "reading the state flag"
      ret = EOS_SW_READFIELD(SWid,"state",state)
   endif

   if arg_present(topog) then begin
      print, "reading the topography"
      ret = EOS_SW_READFIELD(SWid,"topog",topog)
   endif

   print, "done reading - detaching and closing the file"
   ret = EOS_SW_DETACH(SWid)
   ret = EOS_SW_CLOSE(fid)

end


